
Overview of the Mathematica Notebook 
Here we provide a Wolfram Language (Mathematica) [1] implementation of the quantitative 

Donnan-equilibrium analysis used in the accompanying manuscript. The calculations evaluate 

the expected effects of extracellular myo-inositol trispyrophosphate (ITPP) on RBC 

intracellular pH, membrane potential, and cell volume under the assumption that ITPP behaves 

as an extracellular impermeant hexaanion. The theoretical framework is based on the classical 

membrane-equilibrium treatment of Donnan (1911) [2], while the quantitative RBC calibration 

is derived from the measurements of pH gradients and membrane potentials reported by Kirk, 

Kuchel and Labotka (1988) [3]. 

The Notebook first verifies the internal consistency of the Kirk et al. data [3] and then derives 

an empirical relationship between extracellular impermeant charge and intracellular 

alkalinisation. This calibration is subsequently applied to ITPP concentrations used in the 

present work and in previously published studies, including those of Schneider et al., (2019) 

[4]  Duarte et al. (2010) [5], and Koj et al. (2025) [6]. 

Predicted Donnan-induced pH changes are compared with the 31P NMR observations reported 

by Koj et al. [6] and with the oxygen-affinity effects reported by Duarte et al. [5] . The analysis 

further examines whether the reported changes in haemoglobin oxygen affinity might be 

explained by Donnan equilibria alone and quantifies the osmotic consequences of exposing 

RBCs to high extracellular concentrations of ITPP.  

All constants, assumptions, intermediate calculations and graphical outputs are explicitly 

defined, allowing complete transparency and reproducibility of the analysis.  

Summary of Key Findings from the Notebook 

1. The RBC Donnan response is quantitatively validated by the data of Kirk, Kuchel and 

Labotka (1988) [3]. The proton distribution ratios calculated from measured intracellular 

and extracellular pH values closely match the independently measured hypophosphite 

distribution ratios, confirming the applicability of the Donnan framework over membrane 

potentials spanning approximately −8 to +12 mV.  

2. The calibrated Donnan response corresponds to approximately 0.105 pH units per 100 

mEq L⁻¹ of extracellular impermeant anionic charge. Over the experimentally relevant 

range (0-153 mEq L⁻¹), the relationship between extracellular impermeant charge and 

intracellular alkalinisation is approximately linear.  

3. At the ITPP concentration used in the present study (0.9 mM), the predicted Donnan-

induced alkalinisation is negligible. The calculated intracellular pH increase is only about 

+0.006 pH units. 

4. At the ITPP concentration used by Koj et al. [6] (approximately 4.1 mM), the Donnan 

model predicts a small intracellular alkalinisation of about +0.026 pH units.  

5. The direction of the pH change observed by Koj et al. [6] is consistent with the Donnan 

prediction. Analysis of the reported 31P NMR chemical-shift changes yields an estimated 

intracellular alkalinisation of approximately +0.083 pH units, compared with a predicted 

value of +0.026 pH units. Thus, experiment and theory agree in sign and are of the same 

order of magnitude.  

6. If ITPP were completely impermeant, very large extracellular ITPP concentrations would 

be predicted to produce substantial intracellular alkalinisation. Predicted values are 

approximately +0.49 pH units at 78 mM ITPP, +0.75 pH units at 120 mM ITPP, and +1.51 

pH units at 240 mM ITPP. 

7. The 34% increase in P50 reported for 120 mM ITPP cannot be explained by a Donnan 

mechanism alone. The observed right-shift of the oxygen-equilibrium curve requires 

intracellular acidification of approximately 0.26 pH units, whereas the Donnan model 

predicts alkalinisation of approximately +0.75 pH units. The predicted and required pH 

changes are therefore opposite in sign. 

8. An additional acidifying mechanism must therefore operate if the reported oxygen-affinity 

effects are correct. Donnan-induced alkalinisation alone cannot account for the observed 

right-shift of the oxygen-equilibrium curve.  

9. The osmotic consequences of high extracellular ITPP concentrations are predicted to be 

severe. Assuming complete extracellular retention of ITPP, RBCs would retain only about 

44%, 34%, and 20% of their original cell water at 78, 120, and 240 mM ITPP, respectively.  

10. The incubation conditions used in the present work are essentially isotonic. Addition of 

0.9 mM ITPP increases medium osmolality by only ~1.5%, giving a predicted relative cell-

water volume of 0.986.  

11. The Donnan effect is sufficient to explain only modest intracellular pH changes at low-

millimolar ITPP concentrations. At concentrations employed in the present study and by 

Koj et al. [6] predicted pH shifts are small but measurable.  

12. The combination of Donnan and osmotic calculations provides a quantitative benchmark 

against which proposed ITPP transport or intracellular actions can be evaluated. Any 

interpretation of high-concentration ITPP experiments must simultaneously satisfy the 

observed pH changes, oxygen-affinity changes, and osmotic constraints.  

Summary of the Output  
These 12 points move in sequence from validation → calibration → predictions → comparison 

with experiments → physiological implications. 

Quantitative Donnan analysis...ITPP in RBCs
HOW TO RUN 

Open this file in Mathematica (File > Open), or copy the sections  into a new 

Notebook 

Evaluate top to bottom with Shift+Enter   
Section 7 writes the figure as a PDF next to this file   

Every section prints its result. Expected values are given in the 

comments so you can confirm the port is faithful  

In[1]:= ClearAll["Global`*"];

1.  Define parameters and their values

In[2]:=

RTF    = 25.69;    (* RT/F in mV at 298 K *)
zITPP  = 6.;       (* anionic charge on the ITPP hexa-anion *)
mwITPP = 737.8;    (* g/mol, hexasodium salt *)
bohr   = -0.48;    (* d log P50 / d pH, human whole blood *)

teal   = RGBColor[29/255, 158/255, 117/255];
purple = RGBColor[83/255, 74/255, 183/255];
coral  = RGBColor[216/255, 90/255, 48/255];
grey   = RGBColor[136/255, 135/255, 128/255];
red    = RGBColor[163/255, 45/255, 45/255];
blue   = RGBColor[24/255, 95/255, 165/255];
band   = RGBColor[192/255, 221/255, 151/255];

labStyle = Directive[FontFamily -> "Helvetica", FontSize -> 10];



2.  Kirk, Kuchel & Labotka 1988, Biophys J 54:241-247, Table I    
Extracellular NaCl replaced iso-osmotically with Na3-citrate

In[17]:= citrate = {0., 25., 51., 76.}; (* mM, extracellular *)

pHi = {6.97, 7.13, 7.24, 7.41}; (* measured, electrode + NMR*)
pHo = {7.11, 7.19, 7.22, 7.21};
rHPA = {0.65, 0.79, 1.00, 1.59}; (* hypophosphite ratio, Table I *)

Zout = 3. citrate; (* extracellular impermeant anionic charge *)

logr = pHi - pHo; (* = Log10[r] by the Donnan relation *)

rH = 10.^logr;
Em = RTF Log[rH]; (* membrane potential, mV *)

Print[Style["1. Kirk et al. 1988 Table I -- internal consistency", Bold]];
Print[Grid[
Prepend[
Transpose[{citrate, Zout, pHi, pHo,

Round[rH, 0.01], rHPA,
Round[logr, 0.001], Round[Em, 0.1]}],

{"citrate/mM", "Zo/mEq", "pHi", "pHo", "r(H+)", "r(HPA)",
"log r", "Em/mV"}],

Frame  All, Alignment  Right, Spacings  {1.4, 0.8}]];

Print["mean |r(H+) - r(hypophosphite)| = ",
Round[Mean[Abs[rH - rHPA]], 0.001],
" -> the Donnan relation r(Cl) = r(permeant anion) holds"];

Print["Em spans ", Round[First[Em], 0.1], " to ", Round[Last[Em], 0.1],
" mV (Kirk reported -11 to +12 mV)"];

(* EXPECTED
r(H+) 0.72 0.87 1.05 1.58
log r -0.140 -0.060 0.020 0.200
Em/mV -8.3 -3.5 1.2 11.8
mean r(H+) - r(HPA) = 0.052

*)

1. Kirk et al. 1988 Table I -- internal consistency

citrate/mM Zo/mEq pHi pHo r(H+) r(HPA) log r Em/mV

0. 0. 6.97 7.11 0.72 0.65 -0.14 -8.3

25. 75. 7.13 7.19 0.87 0.79 -0.06 -3.5

51. 153. 7.24 7.22 1.05 1. 0.02 1.2

76. 228. 7.41 7.21 1.58 1.59 0.2 11.8

mean |r(H+) - r(hypophosphite)| = 0.052
-> the Donnan relation r(Cl) = r(permeant anion) holds

Em spans -8.3 to 11.8 mV (Kirk reported -11 to +12 mV)

3.  Calibration:  d(pHi) / d(Zout)                    

In[29]:= segSlopes = Differences[logr]/Differences[Zout];

sel     = {1, 2, 3};                      (* Zout <= 153 mEq/L  *)
fitData = Transpose[{Zout[[sel]], logr[[sel]]}];
lm      = LinearModelFit[fitData, z, z];
{cLin, kLin} = lm["BestFitParameters"];

Print[Style["\n2.  Calibration of the Donnan response", Bold]];
Do[Print["  Zout ", Round[Zout[[i]]], " -> ", Round[Zout[[i + 1]]],
         " mEq/L :  d(log r)/dZ = ",
         Round[1000 segSlopes[[i]], 0.001], " x10^-3 per mEq/L"],
   {i, 3}];
Print["  linear coefficient over Zout = 0-153 mEq/L:"];
Print["      kLin = ", Round[1000 kLin, 0.001],
      " x10^-3 pH units per mEq/L"];
Print["           = ", Round[100 kLin, 0.001],
      " pH units per 100 mEq/L, at clamped pHo"];
Print["  NOTE the response accelerates above ~150 mEq/L as [Cl]o becomes"];
Print["  small, so kLin is a LOWER BOUND at high charge loading."];

(* EXPECTED
   segment slopes  1.067, 1.026, 2.400  x10^-3
   kLin = 1.046 x10^-3  ->  0.105 pH units per 100 mEq/L
   cLin = -0.139
*)

2. Calibration of the Donnan response

Zout 0 -> 75 mEq/L : d(log r)/dZ = 1.067 x10^-3 per mEq/L

Zout 75 -> 153 mEq/L : d(log r)/dZ = 1.026 x10^-3 per mEq/L

Zout 153 -> 228 mEq/L : d(log r)/dZ = 2.4 x10^-3 per mEq/L

linear coefficient over Zout = 0-153 mEq/L:

kLin = 1.046 x10^-3 pH units per mEq/L

= 0.105 pH units per 100 mEq/L, at clamped pHo

NOTE the response accelerates above ~150 mEq/L as [Cl]o becomes

small, so kLin is a LOWER BOUND at high charge loading.

 4.  Prediction for the published ITPP protocols      

In[45]:= cKoj = 3.0/mwITPP*1000.;     (* Koj 3 mg/mL -> mM *)

protocols = {
  {"Kuchel (this work)",     0.9},
  {"Koj 2025 in vitro",      cKoj},
  {"Okninska 2020 loading",  78.},
  {"Duarte 2010 OEC",        120.},
  {"Duarte 2010 stock",      240.}};

predTable = Table[
  {p[[1]], Round[p[[2]], 0.01], Round[zITPP p[[2]], 1],
   Round[kLin zITPP p[[2]], 0.001]},
  {p, protocols}];

Print[Style["\n3.  Predicted intracellular alkalinisation, no permeation",
            Bold]];
Print[Grid[Prepend[predTable,
   {"protocol", "[ITPP]/mM", "Zo/mEq", "dpHi predicted"}],
  Frame -> All, Alignment -> {{Left, Right, Right, Right}},
  Spacings -> {1.4, 0.8}]];

(* EXPECTED  dpHi = 0.006, 0.026, 0.489, 0.753, 1.506 *)

3. Predicted intracellular alkalinisation, no permeation

protocol [ITPP]/mM Zo/mEq dpHi predicted

Kuchel (this work) 0.9 5 0.006

Koj 2025 in vitro 4.07 24 0.026

Okninska 2020 loading 78. 468 0.489

Duarte 2010 OEC 120. 720 0.753

Duarte 2010 stock 240. 1440 1.506

4    ITPP Donnan_Analysis_260802.nb



5.  Koj 2025 in vitro: predicted vs observed    

In[55]:= ZKoj      = zITPP cKoj;
dpHpred   = kLin ZKoj;

(* 2,3-DPG 31P chemical shift vs pH, from Koj's own reported values:
   intact RBC 3-P at nominal pH 6.8 -> 3.20 ppm; at pH 7.0 -> 3.32 ppm *)
slopePPM  = (3.32 - 3.20)/(7.0 - 6.8);
dShiftObs = 0.05;                        (* ppm, downfield, Koj Fig 2 *)
dpHobs    = dShiftObs/slopePPM;
dpHobsLo  = dShiftObs/0.9;               (* slope range 0.5 - 0.9 ppm/pH *)
dpHobsHi  = dShiftObs/0.5;

Print[Style["\n4.  Koj 2025 in vitro -- predicted vs observed", Bold]];
Print["  [ITPP] = ", Round[cKoj, 0.01], " mM   ->  Zo = ",
      Round[ZKoj, 0.1], " mEq/L"];
Print["  PREDICTED dpHi (lower bound) = +", Round[dpHpred, 0.001]];
Print["  2,3-DPG shift/pH slope       = ", slopePPM, " ppm per pH unit"];
Print["  OBSERVED  dpHi               = +", Round[dpHobs, 0.001],
      "   (range +", Round[dpHobsLo, 0.001],
      " to +", Round[dpHobsHi, 0.001], ")"];
Print["  ratio observed/predicted     = ", Round[dpHobs/dpHpred, 0.1], "x"];
Print["  SIGN AGREES; magnitude within a factor of 2-4."];
Print["  The prediction is a lower bound because (i) their HEPES buffer"];
Print["  clamps pHo, and (ii) ITPP was added, not iso-osmotically"];
Print["  substituted, so the cells also shrink."];

(* EXPECTED  dpHpred = 0.026 ; dpHobs = 0.083 (range 0.056 - 0.100) *)

4. Koj 2025 in vitro -- predicted vs observed

[ITPP] = 4.07 mM -> Zo = 24.4 mEq/L

PREDICTED dpHi (lower bound) = +0.026

2,3-DPG shift/pH slope = 0.6 ppm per pH unit

OBSERVED dpHi = +0.083 (range +0.056 to +0.1)

ratio observed/predicted = 3.3x

SIGN AGREES; magnitude within a factor of 2-4.

The prediction is a lower bound because (i) their HEPES buffer

clamps pHo, and (ii) ITPP was added, not iso-osmotically

substituted, so the cells also shrink.

6.  Duarte 2010: the 34% P50 right-shift needs ACIDIFICATION 

Ou t [ ] =

-ACIDIFICATION needs shift + 34 Null P50 right the

In[77]:= p50ctrl = 24.25; (* Torr, Duarte Fig 1A control *)

p50itpp = 32.47; (* Torr, Duarte Fig 1A + ITPP *)

dlogP50 = Log10[p50itpp/p50ctrl];
dpHreq = dlogP50/bohr;

Print[Style["\n5. Duarte 2010 -- the P50 right-shift", Bold]];
Print[" P50 ratio = ", Round[p50itpp/p50ctrl, 0.001],

" (", Round[100 (p50itpp/p50ctrl - 1)], "% rise)"];
Print[" d log P50 = +", Round[dlogP50, 0.001]];
Print[" Bohr coefficient = ", bohr];
Print[" => required dpHi = ", Round[dpHreq, 0.001],

" i.e. ACIDIFICATION"];
Print[" Donnan at 120 mM predicts ALKALINISATION of +",

Round[kLin zITPP 120., 0.01], " -> a LEFT shift."];
Print[" Donnan therefore cannot explain the right-shift; an independent"];
Print[" acidifying mechanism is needed (ITPP acid strength)."];
Print[" Bohr range -0.40 to -0.60 gives required dpHi ",

Round[dlogP50/-0.4, 0.01], " to ", Round[dlogP50/-0.6, 0.01],
" -- sign robust."];

(* EXPECTED dlogP50 = 0.127 ; dpHreq = -0.264 *)

5. Duarte 2010 -- the P50 right-shift

P50 ratio = 1.339 (34% rise)

d log P50 = +0.127

Bohr coefficient = -0.48

=> required dpHi = -0.264 i.e. ACIDIFICATION

Donnan at 120 mM predicts ALKALINISATION of +0.75 -> a LEFT shift.

Donnan therefore cannot explain the right-shift; an independent

acidifying mechanism is needed (ITPP acid strength).

Bohr range -0.40 to -0.60 gives required dpHi -0.32 to -0.21 -- sign robust.

 7.  Tonicity of the published incubations 

In[90]:=

phi    = 0.70;    (* osmotic coefficient, polyvalent electrolyte *)
npart  = 7.;      (* ITPP(6-) + 6 Na+ *)
osmIso = 300.;

vrel[c_] := osmIso/(osmIso + phi npart c);

tonTable = Table[
  {Round[c, 0.01], Round[phi npart c], Round[osmIso + phi npart c],
   Round[vrel[c], 0.001], Round[1/vrel[c], 0.01]},
  {c, {0.9, cKoj, 78., 120., 240.}}];

Print[Style["\n6.  Osmotic consequence of the added ITPP", Bold]];
Print[Grid[Prepend[tonTable,
   {"[ITPP]/mM", "osmol added", "total/mOsm", "rel cell water",
    "[Hb] rise"}],
  Frame -> All, Alignment -> Right, Spacings -> {1.4, 0.8}]];
Print["  Duarte's 120 mM incubation was ",
      Round[(osmIso + phi npart 120.)/osmIso, 0.01],
      "x isotonic; cells lose ",
      Round[100 (1 - vrel[120.])], "% of their water."];
Print["  This work at 0.9 mM is ",
      Round[100 ((osmIso + phi npart 0.9)/osmIso - 1), 0.1],
      "% hypertonic -- physiologically innocuous."];

(* EXPECTED  rel cell water 0.986, 0.938, 0.440, 0.338, 0.203 *)

6. Osmotic consequence of the added ITPP

[ITPP]/mM osmol added total/mOsm rel cell water [Hb] rise

0.9 4 304 0.986 1.01

4.07 20 320 0.938 1.07

78. 382 682 0.44 2.27

120. 588 888 0.338 2.96

240. 1176 1476 0.203 4.92

Duarte's 120 mM incubation was 2.96x isotonic; cells lose 66% of their water.

This work at 0.9 mM is 1.5% hypertonic -- physiologically innocuous.

8.  Figure -- three panels 
In[105]:=

(* -- Panel A: calibration, with a second axis in mV -------------------- *)

rightTicks = Table[{v/(RTF Log[10]), v},
  {v, {-7.5, -5, -2.5, 0, 2.5, 5, 7.5, 10, 12.5}}];

panelA = Show[
  Plot[kLin z + cLin, {z, 0, 230},
    PlotStyle -> Directive[purple, Thickness[0.006]]],

  ListPlot[Transpose[{Zout, logr}],
    PlotStyle -> Directive[teal, PointSize[0.03]]],
  Frame -> True,
  FrameLabel -> {
    {"log r    ( = pHi - pHo )", "Em (mV)"},
    {"Extracellular Impermeant Charge  Zo (mEq/L)", None}},
  FrameTicks -> {{Automatic, rightTicks}, {Automatic, None}},
  PlotRange -> {{-8, 240}, {-0.17, 0.225}},
  GridLines -> {None, {{0, Directive[Gray, Dotted]}}},
  PlotLabel -> Style["A    Donnan Calibration (Kirk et al. 1988)",
                     Bold, 14],
  LabelStyle -> labStyle, ImageSize -> 380, AspectRatio -> 0.85];

(* -- Panel B: predicted dpHi vs [ITPP] -------------------------------- *)

markConc = {{0.9, "This Work", teal}, {cKoj, "Koj 2025", coral},
            {78., "Okninska 2020", grey}, {120., "Duarte 2010", red}};

vlinesB = Table[
  ListLogLinearPlot[{{m[[1]], -0.05}, {m[[1]], 1.0}}, Joined -> True,
    PlotStyle -> Directive[m[[3]], Thickness[0.004],
                           Dashing[{0.02, 0.015}]]],
  {m, markConc}];

errBarKoj = ListLogLinearPlot[{{cKoj, dpHobsLo}, {cKoj, dpHobsHi}},
  Joined -> True, PlotStyle -> Directive[coral, Thickness[0.008]]];

panelB = Show[
  LogLinearPlot[kLin zITPP c, {c, 0.1, 250},
    PlotStyle -> Directive[purple, Thickness[0.006]]],
  Sequence @@ vlinesB,
  errBarKoj,
  ListLogLinearPlot[{{cKoj, dpHobs}},
    PlotStyle -> Directive[coral, PointSize[0.035]]],
  Frame -> True,
  FrameLabel -> {{"Predicted dpHi  (Alkalinisation)", None},
                 {"[ITPP] Added to Medium (mM)", None}},
  PlotRange -> {{Log[0.1], Log[250]}, {-0.05, 1.0}},
  PlotLabel -> Style["B     ITPP with no Permeation", Bold, 14],
  Epilog -> Table[
    Text[Style[m[[2]], m[[3]], 8], {Log[m[[1]]], 0.95}, {-1, 0}, {0, 1}],
    {m, markConc}],
  LabelStyle -> labStyle, ImageSize -> 380, AspectRatio -> 0.85];

(* -- Panel C: osmotic consequence ------------------------------------- *)

vlinesC = Table[
  ListLogLinearPlot[{{m[[1]], 0.15}, {m[[1]], 1.12}}, Joined -> True,
    PlotStyle -> Directive[m[[3]], Thickness[0.004],
                           Dashing[{0.02, 0.015}]]],
  {m, markConc}];

panelC = Show[
  LogLinearPlot[vrel[c], {c, 0.1, 250},
    PlotStyle -> Directive[blue, Thickness[0.006]]],

  Sequence @@ vlinesC,
  Frame -> True,
  FrameLabel -> {{"Relative Cell Water Volume", None},
                 {"[ITPP] added to Medium (mM)", None}},
  PlotRange -> {{Log[0.1], Log[250]}, {0.15, 1.12}},
  GridLines -> {None, {{1.0, Directive[Gray, Dotted]}}},
  Prolog -> {band, Opacity[0.45],
             Rectangle[{Log[0.1], 0.9}, {Log[250], 1.1}]},
  Epilog -> {Text[Style["Physiological", Darker[Green], 14],
                  {Log[0.13], 0.945}, {-1, 0}]},
  PlotLabel -> Style["C     Osmotic Consequence", Bold, 14],
  LabelStyle -> labStyle, ImageSize -> 380, AspectRatio -> 0.85];

(* -- assemble and export --------------------------------------------- *)

figure = GraphicsColumn[{panelA, panelB, panelC},ImageSize -> 700,Spacings -> 25]

outDir = Quiet@Check[NotebookDirectory[], Directory[]];
Export[FileNameJoin[{outDir, "donnan_figure_Mathematica.pdf"}], figure];
Print[Style["\nFigure written to ", Bold],
      FileNameJoin[{outDir, "donnan_figure_Mathematica.pdf"}]];

figure

Figure written to
/Users/pkuc2216/Sydney Uni Dropbox/Philip Kuchel/_MISCELLANY/RBC-ITPP

260702/ITPP PAPER/_ITPP NATURE COMMUNICATIONS/donnan_figure_Mathematica.pdf

